Formalisation of Fairness Notions in Assignment Problem

COMP30013 Advanced Studies in Computing

Duc Minh Do

Supervisor: Dr Christine Rizkallah

Declaration and Acknowledgements

I certify that

- This report does not incorporate without acknowledgement any material previously submitted for a degree or diploma in any university; and that to the best of my knowledge and belief it does not contain any material previously published or written by another person where due reference is not made in the text.
- Where necessary I have received clearance for this research from the University's Ethics Committee and have submitted all required data to the School
- The report is 4498 words in length (excluding text in images, table, bibliographies and appendices).

This is my original work, under the supervision of Dr Christine Rizkallah.
I would like to thank Christine for the support and insightful comments.

Contents

List of Figures

Abstract

In the modern-day interconnected world, collaboration across diverse groups has become an inevitable process. With individuals and organizations collaborating more than ever - across cultures, industries, and platforms - ensuring fairness in decision-making is important in creating a trustworthy and equitable collaborative environment. In this context, the need for rigorously defined and enforceable fairness principles should be emphasised, to enforce fairness in our decision processes. This work explores different notions of fairness in social choice theory, particularly in the context of assignment problem, to establish a formal framework for analyzing and implementing these concepts in decision systems. Social choice theory has a rich landscape where competing ideals of fairness can intersect and sometimes conflict, which presents challenges for the design of an equitable decision process. With the aid of formal methods, particularly by the use of proof assistants and the expressiveness of logical frameworks, this study translates notions of fairness into precise mathematical definitions that could ease the process of evaluation and application. In addition to formalising fairness notions, we also develop proofs that highlight the interrelationship among these notions. By mapping out the relationship among these criteria, it enables us to utilise different notions of fairness in various situations, allowing us to consider fairness tailoring to other specific needs.

Chapter 1

Introduction

Social choice theory is an area of study that examines how individual preferences can be aggregated to reach collective decisions [1]. The area itself can be considered a multidisciplinary area, where it combines ideas from other disciplines, such as economics, mathematics, political science, computer science, and philosophy. Social choice theory is concerned with many problems, but they all share the same central question: how to effectively combine each individual's input into results that best reflect all individuals's judgments.

One subarea in social choice theory concerns voting problems, where it focuses on how to aggregate individual preferences over candidates (or ballots) into a single winning candidate, a set of candidates, or an ordering of the candidates [2]. It addresses fundamental questions about different voting systems, examining how various methods can lead to divergent outcomes with the same underlying preferences. This subarea draws on several well-known foundational results in social choice theory, such as Arrow's Impossibility Theorem, the Condorcet Paradox, and Gibbard–Satterthwaite theorem, which demonstrates the challenges of creating a voting system that satisfies a set of seemingly reasonable criteria [3, 4].

Another subarea, which is the main focus of this work - assignment problem, instead concerned about allocating objects, and resources to individuals given the preferences of the agents. The area is also known as the "house allocation problem", "two-sided matching with one-sided preferences" [5]. In this subarea, the main concern is to allocate limited resources to a number of given agents, according to their preferences on the resources. There are several scenarios to which the problem is concerned. When the resource is divisible among the agents, the problem can be framed as a fractional assignment problem, where each agent receives a fraction of each of the resources or alternatives. When the resource is indivisible among the agents, the problem can be tackled with the introduction of either discrete or random assignment. Discrete assignment refers to the situation where each of the agents receives one or more discrete alternatives. Random assignment, meanwhile, refers to the situation where each of the agents is assigned a probability of obtaining the alternatives. This work suggests a way of formalising the setting of all the scenarios of the assignment problem in social choice theory.

The study of fairness in the context of the assignment problem is relevant in multiple scenarios such as school admissions, job placements, public resource distribution, and workplace allocation, where equitable access is essential. This work's main objective is to explore the fair assignment of objects, with the focus on random assignment and discrete assignment problem. We consider two concepts of fairness to formalise for this setting, which are *proportionality* and *envy-freeness*. While proportionality requires that each of the agent should get an allocation that is at least 1/n of the total valuation of the alternatives for all the alternatives (n is the number of agents), envy-freeness requires that no agent value another agent's allocation more than their own allocation. We formulate the notions of envy-freeness and proportionality based on the usage of stochastic dominance (SD) relation, which describes the relationships between allocations, and the usage of utility functions. As for discrete assignment problem, responsive set extension (RS) relation is also introduced as a relation between sets of alternatives and formalised according to the definition in [6]. Different notions of proportionality and envy-freeness will be used as our definitions for fairness notions in assignment problem. This includes *SD proportional*, *weak SD proportional*, *SD envyfreeness* and *weak SD envyfreeness*, the strong and weak notions for each of the concepts proportionality and envyfreeness. The study of a range of fairness criteria allows us to weaken the fairness requirements depending on other contexts.

We will present a theoretical foundation of the assignment problem, alongside the definition of the fairness notions in pen-and-paper notations, then followed by our formalisation framework within Isabelle/HOL.

Chapter 2

Background and Related Works

2.1 Assignment Problem

The assignment problem is a fundamental challenge, not just in social choice theory, but also part of the area of operations research. The problem is also regarded as an allocation problem, where a number of objects are allocated among a number of agents [7, 8]. The simplest setting for this, and the most widely used assumption is that the number of objects is equal to the number of agents, thus each agent will be expected to receive exactly one object [5, 9]. Another assumption in the context of the assignment problem is that each agent has their own preference over the objects. Early works on analysing properties of the assignment problem only concerned with strict preferences, where an agent can only strictly prefer an alternative over another [9]. This assumption raised the concern for impracticality, as an individual when exposed to a set of objects might have indifference preference toward some groups of objects, where they do not necessarily strictly prefer any object over another in the group. With this in mind, several works emerged with a new assumption of indifference in agents' preference over alternatives, which adds more complexity to the model and to the theorems related to the assignment problem [10]. Having had the set of agents, the set of alternatives, and the preference of agents over the alternatives, the goal now of the assignment problem is to assign to each pair of agent and alternative a probability of the agent receiving the object. The unanimous assumption here is that for each alternative, the sum of probability over all agents is 1 [11, 12, 13].

2.2 Fairness Notions in Assignment Problem

Fairness is a central topic to many subareas within social choice theory, and the assignment problem is not an exception. When considering fairness in the context of the assignment problem, two concepts mostly used to describe the fairness of an assignment are *proportionality* and *envy-freeness*. An assignment that satisfies proportionality when all agents believe that they have received their fair share of all the items based on their valuation. Meanwhile, envy-freeness emphasizes that every agent prefers their own share to the share of any other agent, based on their valuation and judgements. We adopt the notion of *SD proportional*, *weak SD proportional*, *SD envyfreeness* and *weak SD envyfreeness* from the article [6]. The notions used in the above-mentioned article involved the use of the stochastic dominance relation, which is usually used to describe the relationship between pairs of lotteries. The concept of SD envy-freeness had been used in several works, but mostly on random assignment [9]. The notion of SD envyfree and weak SD envyfree used in this work is analogous to the notion of *envy-free* and *not envy-ensuring* in [14]. Meanwhile, the notion of SD proportionality is used in [15], with the exception that the setting in the article involves the assumption of strict preference over alternatives.

2.3 Computer-Aided Theorem Proving in Social Choice Theory

The fact that social choice theory is built on the axiomatic foundation with many theorems on the impossibility of scenarios has encouraged the use of computer-aided theorem proving techniques [16]. Many of the works are done towards the formalising and verifying of existing theorems [17, 18, 19, 20]. At the same time, there are also works with the help of computers and theorem provers had devised new results and theorems [21, 22]. Isabelle/HOL is popular among the community of computational social choice as a tool for formalising different concepts in social choice theory. Supporting higher order logic, Isabelle/HOL provides a framework that eases the process of formalisaton of axiomatic foundations in social choice theory. Many of the results currently on Archive of Formal Proofs[1] (AFP) (a collection of proof libraries, examples, and scientific developments that are mechanically checked in the theorem prover Isabelle) mainly concerns about the problems related to the area of voting and characterisations of some impossibility results in this area [19, 22, 23, 24]. In the archive, not only impossibility results in voting theory are formalised, but there is also an effort to formalising the foundational concepts in social choice, which would aid in the formalisation of more complex theorems and models. These works focus on the formalisation of concepts that are fundamental to the field of social choice theory, such as preference profiles, stochastic dominance, and utility functions [25, 26]. These existing formalisations of the foundational concepts in social choice are helpful

[1] www.isa-afp.org/

in the sense that they would allow new formalisations to be constructed more conveniently, under the assumption of these formalisations. In our work, we adapt some definitions of the foundational concepts in the formalisations and use the properties realised in these works to help prove the relationship between fairness concepts. Most of our work is based on the formalisation of stochastic dominance and the formalisation of utility function used in [25].

Chapter 3

Methodology and Results

This chapter discusses our process of formalising the notions of fairness in the assignment problem. We will cover the pen-and-paper definitions for the assignment problem, definition of some relations, and the informal definition for fairness concepts that we are using in this work. We will then follow with some equivalence and implication results among these notions. We then provide a formalisation framework for assignment problem, and the sketch of the proof for some of the equivalences and implication results.

3.1 Preliminaries

3.1.1 Assignment Problem

The assignment problem is characterised by a finite set of **n** agents N, and a finite set of **m** alternatives O. Each agent **i** has their own preference over the set of alternatives, which we denote as $\succeq_i$. Each of the preferences is total and transitive over the set of alternatives O.
Given an assignment p, the allocation of agent i is p_i, and the probability that agent i receives the alternative j is $p_i(j)$. We always have $0 \leq p_i(j) \leq 1$ for any agent i and alternative j.
Also, in order to characterise the sum of probability over all agents given an alternative, we should have $\Sigma_{i \in N} p_i(j) = 1$ for any alternative j.
In the discrete setting, the probabilities can only take two values, either 0 or 1, representing the fact that either the agent has the whole object or none of the objects.
The usual setting for this is that p_i is a lottery for every agent $i \in N$. In other words, it means that the usual setting for the random assignment problem is that $\sum_{j \in O} p_i(j) = 1$ for each agent. However, we found out that the setting might not provide a thorough description of the problem of discrete assignment, as an agent can receive more than one alternative in the discrete setting (if the sum of probability is 1, then there is only one object allocated to each agent). We then decided to give an extension to the usual model. In our new model,

there is no restriction for the sum of probability for each agent to be equal to 1. We only assume that the sum of probability for each agent has to be constant for all agents.

3.1.2 Some Relations on Allocations

There are three ways in which we are using to describe the relation between allocations. The first two are characterised under the assumption of random assignment and the last one is characterised in the scenario of discrete assignment.

Stochastic Dominance

Stochastic dominance usually describes the relationship between lotteries or the relationship between probability mass function, denoted as $\succeq_i^{SD}$ for an agent i (the relation is dependent on the preference over alternatives of the agent). The setting is different since we are not dealing with lotteries anymore. However, the definition of stochastic dominance still works with this new setting, despite the change of assumption. In its usual definition, an agent i "SD prefers" a lottery p_i over a lottery q_i if and only if for each object o, p gives the agent at least as many objects that are at least as preferred as o as q [6]. The definition for this is:
$p_i \succeq_i^{SD} q_i$ if and only if $\sum_{o_j \in o_k : o_k \succeq_i o} p_i(o_j) \geq \sum_{o_j \in o_k : o_k \succeq_i o} q_i(o_j)$ for any $i \in N$ and any $o \in O$
This definition also works with the new setting, as there is no part of definition that requires the sum of probability to be 1.

Utility Functions

Utility functions are used for describing the agent's valuation of the alternatives, which is a function from the set of alternatives to the set of real numbers. A utility function for an agent is a function that satisfies, or that is consistent with the preference profile of that agent.
For an agent and their utility function u_i, this means that for any two alternatives x and y, $x \succeq_i y$ if and only if $u_i(x) \geq u_i(y)$. Though not considered a relation, the definition of utility function also helps to describe the relationship between allocations. Given an allocation for agent i, which is p_i, we use the notion to describe the utility of the whole allocation $u_i(p_i)$, where $u_i(p_i) = \sum_{j \in O} u_i(j) * p_i(j)$
There is a well-known and proven fact that relates stochastic dominance to utility functions. The fact states that $p_i \succeq_i^{SD} q_i$ if and only if for any utility function u_i consistent with the preference relation $\succeq_i$, $u_i(p_i) \geq u_i(q_i)$. There have been several proofs for this fact, all of which rely on the assumption of the sum of probabilities being 1 for a particular agent. In this work, proof for this fact in the new setting, where the sums of probabilities are constants, is

introduced and discussed.
For an utility function, u we denote $u(O) = \sum_{i \in O} u(i)$.

Responsive Set Extension

The two above notations can deal with both random assignment and discrete assignment. However, in the case of discrete assignment, another relation is introduced to capture the discrete nature of the scenario. The Responsive Set Extension (RS) extends the usual relationship between alternatives, instead, it is concerned with the relationship between sets of alternatives [6, 27]. For each agent i in the discrete setting, the allocation p_i can be described as a set of alternatives, instead of a list of probabilities. The definition for responsive set extension states that for two allocations (set of alternative) p and q, we have $p \succeq_i^{RS} q$ if and only if there is an injection f from q to p such that for each $o \in q$, $f(o) \succeq_i o$

3.1.3 Fairness Notions

In this work, we consider some fairness notions, which incorporate the use of the above-mentioned relations. All of our fairness notations are adopted from the paper [6].

Proportionality

SD proportionality An assignment p satisfies *SD proportionality* if each agent SD prefers is allocation to the uniform allocation (n here is the number of agents):
$p_i \succeq_i^{SD} (1/n, ..., 1/n)$ for all $i \in N$

Necessary proportionality An assignment satisfies necessary proportionality if it is proportional for all cardinal utilities consistent with the agents' preference:
For each $i \in N$, and for each utility u_i consistent with $\succeq_i$, such that $u_i(p_i) \geq u_i(O)/n$

Weak SD proportionality An assignment p satisfies *Weak SD proportionality* if no agent strictly SD prefers the uniform allocation to his:
$\neg((1/n, ..., 1/n) \succ_i^{SD} p_i)$ for all $i \in N$

Possible proportionality An assignment satisfies *Possible proportionality* if, for each agent, there exists a cardinal utility function that is consistent with their preferences, and their allocation yields them at least as much the utility as he would get under the uniform allocation:
For each $i \in N$, there exists u_i consistent with $\succeq_i$, such that $u_i(p_i) \geq u_i(O)/n$

Envy-freeness

SD envy-freeness An assignment p satisfies SD envy-freeness if each agent SD prefers their allocation to that of any other agent:
$p_i \succeq_i^{SD} p_j$ for all $i, j \in N$

Necessary envy-freeness An assignment satisfies necessary envy-freeness if, for all cardinal utility functions, any agent values their own allocation at least as much as any other agent's.
For each agent i, and each u_i consistent with $\succeq_i$, then $u_i(p_i) \geq u_i(p_j)$

Necessary completion envy-freeness An assignment satisfies necessary completion envy-freeness if for any agent, each total order over the set of all sets of alternatives that is consistent with RS, each agent weakly prefers their allocation to any others'.

Weak SD envy-freeness An assignment p satisfies *weak SD envy-freeness* if no agent strictly SD prefers other agents' allocation to theirs:
$\neg(p_j \succ_i^{SD} p_i)$ for all $i, j \in N$

Possible envy-freeness An assignment p satisfies *possible envy-freeness* if, for each agent, there are utility functions consistent with their preferences with which they value their allocation at least as much as any other agent's.
For all $i \in N$, there exists u_i consistent with $\succeq_i$ such that $u_i(p_i) \geq u_i(p_j)$ for all $j \in N$

Possible completion envy-freeness An assignment satisfies *possible completion envy-freeness* if for each agent i, there exists a preference relation of the agent over sets of objects that is a weak order consistent with the responsive set extension such that the agent weakly prefers their allocation over the allocations of any other agents.

3.1.4 Relationship between fairness notions

With the above-mentioned fairness notions, we also proved some relationships between these notions.

Equivalences

Each pair (or triple) of notions mentioned below are equivalent to each other.

1. Weak SD proportionality and possible proportionality
2. SD proportionality and necessary proportionality
3. Weak SD envy-freeness and possible completion envy-freeness

4. SD envy-freeness, necessary envy-freeness, and necessary completion envy-freeness

Implications

1. SD envy-freeness implies SD proportionality
2. SD proportionality implies weak SD proportionality
3. Possible envy-freeness implies weak SD proportionality
4. Possible envy-freeness implies weak SD envy-freeness

3.2 Formalisation

3.2.1 Assignment Problem

In the context of formalisation for the assignment problem, to the best of my knowledge, there has not been any available work in this specific area. However, there has been an impressive line of work on formalising the foundational model of social choice theory. The work by Manuel Eberl - Randomised Social Choice [25]- provided a formalisation of several crucial concepts in social choice theory, which provides a solid foundation for our work to build upon and formalise the setting of random and discrete assignment. Two of the concepts used in the work are also reused in my work, which are **preference profile** and **utility function**. Both of them use the assumption that the number of alternatives is finite, and the preference of agents over the alternatives is a total preorder (total, reflexive, and transitive) relation on the set of alternatives. Each property of the preference relation is reasonable for an individual's preference over objects.

The code at 3.1 describes the formalisation of the preference profile used in the AFP session, while the code at 3.2 describes the formalisation of the utility function that is used in our formalisation.

```
locale pref_profile_wf =
  fixes agents :: "'agent set" and alts :: "'alt set" and R :: "('agent, 'alt) pref_profile"
  assumes nonempty_agents [simp]: "agents ≠ {}" and nonempty_alts [simp]: "alts ≠ {}"
  assumes prefs_wf [simp]: "i ∈ agents ⟹ finite_total_preorder_on alts (R i)"
  assumes prefs_undefined [simp]: "i ∉ agents ⟹ ¬R i x y"
begin
```

Figure 3.1: Formalisation of the foundation setting for social choice theory

```
locale vnm_utility = finite_total_preorder_on +
  fixes u :: "'a ⇒ real"
  assumes utility_le_iff: "x ∈ carrier ⟹ y ∈ carrier ⟹ u x ≤ u y ⟷ x ≼[le] y"
begin
```

Figure 3.2: Formalisation of the von Neumann-Morgenstern utility function

Incorporating the formalisation of preference profile as the foundation setting for random and discrete assignment, we first propose the idea of *random allocation*. Although the work in [25] refers to an allocation as a probability mass function (pmf) of the alternatives, we want to extend this notion so that the sum of probability over all alternatives of an agent is not necessarily 1, but sum to a constant that is the same for all agents. Our formalisation for random allocation can be found in figure 3.3. Here, we described the allocation not as a pmf, but rather a function from the set of alternatives to the set of real numbers. With the formalisation of random allocation, we formalised the setting for random assignment as in figure 3.4. We also used the type of *mult_ assignment* to describe a (possibly) multiple assignment, where the number of alternatives can be more than the number of agents.

```
locale random_allocation =
  fixes alts :: "'b set"
  fixes h :: "'b allocation"
  assumes finite_alts [simp]: "finite alts"
  assumes undefined_alts [simp]: "j ∉ alts ⟶ h j = 0"
  assumes prob [simp]: "∀j ∈ alts. (0 ≤ h j ∧ h j ≤ 1)"
```

Figure 3.3: Formalisation of the random allocation concept

```
locale random_assignment =
  fixes agents :: "'agent set" and alts :: "'alt set" and R :: "('agent, 'alt) pref_profile"
  assumes "pref_profile_wf agents alts R"
  fixes p :: "('agent, 'alt) mult_assignment"
  assumes random_alloc [simp]: "∀i ∈ agents. random_allocation alts (p i)"
  assumes undefined_agent [simp]: "∀j. i ∉ agents ⟶ p i j = 0"
  assumes stochastic [simp]: "∀j ∈ alts. (∑ i∈agents. p i j) = 1"
  assumes sum_prob_equal_agents [simp]: "∃c :: real. ∀ i ∈ agents. (∑ j ∈ alts. p i j) = c"
```

Figure 3.4: Formalisation of the random assignment

The concept of discrete allocation and discrete assignment are also formalised, with one difference being that the probabilities can now only be 0 or 1, as in figure 3.5 and 3.6.

```
locale discrete_allocation = random_allocation +
  assumes disc: "∀i ∈ alts. h i = 0 ∨ h i = 1"
```

Figure 3.5: Formalisation of the discrete allocation concept

```
locale discrete_assignment = random_assignment +
  assumes discr: "∀i :: 'agent ∈ agent. discrete_allocation alts ((p :: ('agent, 'alt) mult_assignment) i)"
```

Figure 3.6: Formalisation of the discrete assignment

3.2.2 Relations on Allocations

With the formalisation of the setting, we now formalise the stochastic dominance relations and the responsive set extension relations. Note that the concept of the utility function is formalised in [25] and it fits the current assumptions in this work.

Stochastic Dominance

Our formalisation of stochastic dominance captures the fact that the allocations do not need to be PMFs, as seen in figure 3.7. The inclusion of the relation Ryy in each set ensures that y is actually an object from the set of alternatives.

```
definition SDA :: "'alt relation ⇒ 'alt allocation relation"
  where
  "p ≿[SDA(R)] q ≡ ∀x. (R x x ⟶ (sum p {y. R y y ∧ y ≿[R] x} ≥ sum q {y. R y y ∧ y ≿[R] x}))"
```

Figure 3.7: Formalisation of stochastic dominance

Responsive Set Extension

We also included a formalisation of the Responsive Set Extension relation in the formalisation as in figure 3.8.

```
definition
  RS :: "'alt relation ⇒ 'alt set relation"
where
  "p ≽[RS(P)] q ≡
    ∃f :: ('alt ⇒ 'alt).
      (inj_on f q  ∧ f ` q ⊆ p ∧
      (∀(x:: 'alt) ∈ q. f x ≽[P] x))"
```

Figure 3.8: Formalisation of the Responsive Set Extension

3.2.3 Fairness notions

The fairness notions mentioned in subsection 3.1.3 are formalised in different contexts. While the notions in figure 3.9 and 3.10 work for both scenarios: random and discrete assignment, the notions in figure 3.11 are only applicable to the case of discrete assignment.

```
(* Proportionality *)
definition SD_proportional :: "('agent, 'alt) mult_assignment ⇒ bool"
  where
  "SD_proportional A ≡
    ∀i ∈ agents. (A i) ≽[SDA(R i)] (pmf_like_set alts)"

definition weak_SD_proportional :: "('agent, 'alt) mult_assignment ⇒ bool"
  where
  "weak_SD_proportional A ≡
    ∀i ∈ agents. ¬((pmf_like_set alts) ≻[SDA(R i)] (A i))"

definition possible_proportional :: "('agent, 'alt) mult_assignment ⇒ bool"
  where
  "possible_proportional A ≡
    ∀i ∈ agents. ∃u :: ('alt ⇒ real). vnm_utility alts (R i) u
      ∧ sum_utility u alts (A i) ≥ sum u alts / (card alts)"

definition necessary_proportional :: "('agent, 'alt) mult_assignment ⇒ bool"
  where
  "necessary_proportional A ≡
    ∀i ∈ agents. ∀u :: ('alt ⇒ real).( vnm_utility alts (R i) u
    ⟶ sum_utility u alts (A i)  ≥ sum u alts / (card alts))"
```

Figure 3.9: Formalisation of the proportionality-related notions

```
(* Envy-freeness *)
definition SD_envyfree :: "('agent, 'alt) mult_assignment ⇒ bool"
  where
  "SD_envyfree A ≡
    ∀i ∈ agents. ∀j ∈ agents. (A i) ≿[SDA(R i)] (A j)"

definition weak_SD_envyfree :: "('agent, 'alt) mult_assignment ⇒ bool"
  where
  "weak_SD_envyfree A ≡
    ∀i ∈ agents. ∀j ∈ agents. ¬ ((A j) ≻[SDA(R i)] (A i))"

definition possible_envyfree :: "('agent, 'alt) mult_assignment ⇒ bool"
  where
  "possible_envyfree A ≡
    ∀i ∈ agents. ∃u :: ('alt ⇒ real). ∀j ∈ agents. vnm_utility alts (R i) u
    ∧ sum_utility u alts (A i) ≥ sum_utility u alts (A j)"

definition necessary_envyfree :: "('agent, 'alt) mult_assignment ⇒ bool"
  where
  "necessary_envyfree A ≡
    ∀i ∈ agents. ∀j ∈ agents. ∀u :: ('alt ⇒ real). vnm_utility alts (R i) u
    ⟶ sum_utility u alts (A i) ≥ sum_utility u alts (A j)"
```

Figure 3.10: Formalisation of the envy-freeness-related notions

```
abbreviation RS :: "('alt ⇒ 'alt ⇒ bool) ⇒ 'alt set ⇒ 'alt set ⇒ bool" where
"RS x ≡ pref_profile_wf.RS x"

definition possible_completion_envyfree :: "('agent, 'alt) mult_assignment ⇒ bool"
  where
  "possible_completion_envyfree A ≡
  ∀i ∈ agents. (∃P :: 'alt set ⇒ 'alt set ⇒ bool. (∀s1 s2.  s1 ≿[P] s2 ⟷ s1 ≿[RS(R i)] s2) ∧
  (∀j ∈ agents. allocated_alts (p i) alts ≿[P] allocated_alts (p j) alts))"

definition necessary_completion_envyfree :: "('agent, 'alt) mult_assignment ⇒ bool"
  where
  "necessary_completion_envyfree A ≡
  ∀i ∈ agents. (∀P :: 'alt set ⇒ 'alt set ⇒ bool. (∀s1 s2.  s1 ≿[P] s2 ⟷ s1 ≿[RS(R i)] s2) ⟶
    (∀j ∈ agents. allocated_alts (A i) alts ≿[P] allocated_alts (A j) alts))"
```

Figure 3.11: Formalisation of the discrete only fairness notions

3.2.4 Relationships between fairness notions

The relationships between these fairness notions mentioned in subsection 3.1.4 can be summarised in the figure 3.12, which was obtained from the article [6].



Figure 3.12: Relationships between the fairness notions, the arrows are for implications and the rest are for equivalences, EF is for envyfree, Prop is for proportional, PC is for possible completion, and NC is for necessary completion [6]

In order to prove these relations, we resort to the equivalence among the relations between allocations. A classical result, mentioned in 3.1.2 is formalised as in figure 3.13.

```
locale random_pair_allocation =
  fixes alts :: "'alt set"
  fixes p q :: "'alt allocation"
  fixes R :: "'alt relation"
  assumes nonempty_a: "alts ≠ {}"
  assumes alts_rel: "finite_total_preorder_on alts R"
  assumes ra_p: "random_allocation alts p"
  assumes ra_q: "random_allocation alts q"
  assumes sum_prob: "(∑i ∈ alts. p i) = (∑i ∈ alts. q i)"
begin

lemma frac_SDA_utility:
  "p ≥[SDA(R)] q ⟷ (∀u. vnm_utility alts R u ⟶ sum_utility u alts p ≥ sum_utility u alts q)"
```

Figure 3.13: Formalisation of the relationship between stochastic dominance and utility functions

We will provide a sketch of the proof for this lemma, which heavily relies on the definition in 3.2.2.

For the left-to-right direction, we prove that if there is a total, reflexive, transitive relation on a finite, nonempty set, there must be a way to arrange the elements of that set in a sequence such that the order of the elements is consistent with the given relation. From this lemma, combining with the technique of Abel's summation, and the fact that an utility function is a monotonic function on this sequence, we should obtain the desired result.

For the right-to-left direction, given an arbitrary alternative x we want to choose a ϵ-family of utility function that is in the form $a + b * \epsilon$, such that the function

a is the indicator function for the set $\{y|y \succeq x\}$, and b is an utility function (the fact that $a + b * \epsilon$ is also an utility function is also proven in [25]). Use the right-hand side of the lemma and take ϵ sufficiently small, we should obtain the desired result.

Also, equivalence in the discrete setting is proven as a lemma as in figure 3.14 and 3.15

```
lemma RS_utility:
  "allocated_alts p alts ≿[RS(R)] allocated_alts q alts ⟷
     (∀u. vnm_utility alts R u
         ⟶ sum u (allocated_alts p alts) ≥ sum u (allocated_alts q alts))"
```

Figure 3.14: Formalisation of the relationship between responsive set extension and utility functions

```
lemma SDA_RS:
  "p  ≿[SDA(R)] q  ⟷
   allocated_alts p alts ≿[RS(R)] allocated_alts q alts"
```

Figure 3.15: Formalisation of the relationship between SD and RS

To prove this two equivalences, we rely on the result in 3.13.
We first provide a sketch proof for a weaker result, illustrated in 3.16. The proof for this is also mentioned in [6] which relies on graph theory, particularly on bipartite graph, with the usage of Hall's theorem. Also, the proof itself is by contradiction. We assume the contrary, and assume a bipartite graph with one set of vertices be the set of alternatives in one allocation, and another set of vertices be the set of alternatives in the other allocation. With the assumption, we obtain that the bipartite graph does not have a matching saturating one set of alternatives. Using Hall's Theorem, and assume that the set mentioned in Hall's Theorem is maximal, from which we can obtain the result.

```
lemma SDA_to_RS:
  "p ≿[SDA(R)] q   ⟶
   allocated_alts p alts ≿[RS(R)] allocated_alts q alts"
```

Figure 3.16: Formalisation of the implication relationship between SD and RS

A second result to prove to show that all three relations are equivalent in the case for discrete setting is a weaker result of 3.14, illustrated in 3.17. The proof for the second result is trivial, with only one reminder that under the discrete assignment setting, the utility of an allocation is just the utility summing over the set of objects or alternatives in that allocation.

```
lemma RS_to_utility:
  "allocated_alts p alts ≿[RS(R)] allocated_alts q alts ⟶
     (∀u. vnm_utility alts R u
         ⟶ sum u (allocated_alts p alts) ≥ sum u (allocated_alts q alts)) "
```

Figure 3.17: Formalisation of the implication relationship between RS and utility functions

Combining the results in 3.13, 3.16, and 3.17, we can conclude that the three relations are equivalent under the discrete assignment setting.

All of the definitions of the fairness notions mentioned above are defined using the three relations above, and having the equivalences among the three will conclude the relationship between the fairness notions.

Chapter 4

Conclusion

In this research project, a formal framework for the assignment problem along with various fairness notions was developed in Isabelle/HOL. The formalisation not only rigorously defines the setting of the assignment problem and introduces some notions of fairness in this setting, but also enables a thorough analysis of these criterias. The use of Isabelle/HOL has allowed us to ensure that our definitions and theorems are precise and verifiable. Through this work, we contribute to the broader field of social choice theory by providing a formal framework that could provide a foundation for future research on the assignment problem.

Regarding future works, there are many open directions available in the area of assignment problem in the intersection with formal methods. In our work, the current focus is merely on providing a formal theoretical framework for the random and discrete assignment problem and the properties regarding fairness concepts. An extension to this within the area of formal methods is to analyse the actual algorithmic complexity for verifying if a certain assignment satisfy some certain fairness criteria. Another direction to look at is the extension of the framework of the assignment problem to incorporate other concepts such as strategyproofness, which is concerned about the incentive of agents in the allocation process, or efficiency, which aims to maximise overall welfare for the agents. Another suggestion is a formalisation of a more general framework for the assignment problem. Our current framework underpins an important assumption, which is that the sum of probabilities of an agent receiving an object, over all objects, is constant for all agents in the problem. An ideal generalised framework is a framework that would not restrict the sum of probabilities of an agent receiving an object over all objects, which should be more reasonable for an assignment or allocation setting. This framework's only assumption is that the probability of receiving an object, summing over all agents, is 1.

Bibliography

[1] Christian List. Social Choice Theory. In Edward N. Zalta and Uri Nodelman, editors, *The Stanford Encyclopedia of Philosophy*. Metaphysics Research Lab, Stanford University, Winter 2022 edition, 2022.

[2] Eric Pacuit. Voting Methods. In Edward N. Zalta and Uri Nodelman, editors, *The Stanford Encyclopedia of Philosophy*. Metaphysics Research Lab, Stanford University, Summer 2024 edition, 2024.

[3] Kenneth J. Arrow. *Social Choice and Individual Values*. Yale University Press, 2012.

[4] Mark Allen Satterthwaite. Strategy-proofness and arrow's conditions: Existence and correspondence theorems for voting procedures and social welfare functions. *Journal of Economic Theory*, 10(2):187–217, 1975.

[5] Haris Aziz, Felix Brandt, and Paul Stursberg. On popular random assignments. In Berthold Vöcking, editor, *Algorithmic Game Theory*, pages 183–194, Berlin, Heidelberg, 2013. Springer Berlin Heidelberg.

[6] Haris Aziz, Serge Gaspers, Simon Mackenzie, and Toby Walsh. Fair assignment of indivisible objects under ordinal preferences. *Artificial Intelligence*, 227:71–92, 2015.

[7] David J. Abraham, Katarína Cechlárová, David F. Manlove, and Kurt Mehlhorn. Pareto optimality in house allocation problems. In Rudolf Fleischer and Gerhard Trippen, editors, *Algorithms and Computation*, pages 3–15, Berlin, Heidelberg, 2005. Springer Berlin Heidelberg.

[8] Peter Gardenfors. Assignment problem based on ordinal preferences. *Management Science*, 20(3):331–340, 1973.

[9] Anna Bogomolnaia and Hervé Moulin. A new solution to the random assignment problem. *Journal of Economic Theory*, 100(2):295–328, 2001.

[10] Akshay-Kumar Katta and Jay Sethuraman. A solution to the random assignment problem on the full preference domain. *Journal of Economic Theory*, 131(1):231–250, 2006.

[11] Haris Aziz and Yoichi Kasajima. Impossibilities for probabilistic assignment. *Social Choice and Welfare*, 49(2):255–275, August 2017. Funding Information: Two independent papers, Aziz () and Kasajima (), were merged into the current paper. Kasajima () was based on a chapter of his Ph.D. thesis at the University of Rochester. Aziz is supported by a Julius Career Award. Kasajima acknowledges support from the JSPS KAKENHI Grant Numbers 22830102 and 16K03561. Publisher Copyright: © 2017, Springer-Verlag Berlin Heidelberg.

[12] Fuhito Kojima. Random assignment of multiple indivisible objects. *Mathematical Social Sciences*, 57(1):134–142, 2009.

[13] Haris Aziz, Rupert Freeman, Nisarg Shah, and Rohit Vaish. Best of both worlds: Ex ante and ex post fairness in resource allocation. *Operations Research*, 72(4):1674–1688, 2024.

[14] Steven J Brams, Paul H Edelman, and Peter C Fishburn. Fair division of indivisible items. *Theory and Decision*, 55:147–180, 2003.

[15] Kirk Pruhs and Gerhard J Woeginger. Divorcing made easy. In *Fun with Algorithms: 6th International Conference, FUN 2012, Venice, Italy, June 4-6, 2012. Proceedings 6*, pages 305–314. Springer, 2012.

[16] Haris Aziz, Felix Brandt, Edith Elkind, and Piotr Skowron. Computational social choice: The first ten years and beyond. In Bernhard Steffen and Gerhard J. Woeginger, editors, *Computing and Software Science - State of the Art and Perspectives*, volume 10000 of *Lecture Notes in Computer Science*, pages 48–65. Springer, 2019.

[17] Peter Gammie. Some classical results in social choice theory. *Archive of Formal Proofs*, November 2008. `https://isa-afp.org/entries/SenSocialChoice.html`, Formal proof development.

[18] Kwing Hei Li. Formalizing may's theorem. *CoRR*, abs/2210.05342, 2022.

[19] Tobias Nipkow. Arrow and gibbard-satterthwaite. *Archive of Formal Proofs*, September 2008. `https://isa-afp.org/entries/ArrowImpossibilityGS.html`, Formal proof development.

[20] Wesley H. Holliday, Chase Norman, and Eric Pacuit. Voting theory in the lean theorem prover. In Sujata Ghosh and Thomas Icard, editors, *Logic, Rationality, and Interaction*, pages 111–127, Cham, 2021. Springer International Publishing.

[21] Felix Brandt and Christian Geist. Finding strategyproof social choice functions via sat solving. In *Proceedings of the 2014 International Conference on Autonomous Agents and Multi-Agent Systems*, AAMAS '14, page 1193–1200, Richland, SC, 2014. International Foundation for Autonomous Agents and Multiagent Systems.

[22] Felix Brandt, Manuel Eberl, Christian Saile, and Christian Stricker. The incompatibility of fishburn-strategyproofness and pareto-efficiency. *Archive of Formal Proofs*, March 2018. `https://isa-afp.org/entries/Fishburn_Impossibility.html`, Formal proof development.

[23] Manuel Eberl. The incompatibility of sd-efficiency and sd-strategy-proofness. *Archive of Formal Proofs*, May 2016. `https://isa-afp.org/entries/SDS_Impossibility.html`, Formal proof development.

[24] Théo Delemazure, Tom Demeulemeester, Manuel Eberl, Jonas Israel, and Patrick Lederer. The incompatibility of strategy-proofness and representation in party-approval multi-winner elections. *Archive of Formal Proofs*, November 2022. `https://isa-afp.org/entries/PAPP_Impossibility.html`, Formal proof development.

[25] Manuel Eberl. Randomised social choice theory. *Archive of Formal Proofs*, May 2016. `https://isa-afp.org/entries/Randomised_Social_Choice.html`, Formal proof development.

[26] Julian Parsert and Cezary Kaliszyk. Von-neumann-morgenstern utility theorem. *Archive of Formal Proofs*, July 2018. `https://isa-afp.org/entries/Neumann_Morgenstern_Utility.html`, Formal proof development.

[27] Salvador Barberà, Walter Bossert, and Prasanta K Pattanaik. Ranking sets of objects. In *Handbook of Utility Theory: Volume 2 Extensions*, pages 893–977. Springer, 2004.